Appendix B

Coated Sphere

As might be expected, adding a coating to a homogeneous sphere leads to several new computational problems, not all of which we were able to completely solve. As a consequence, BHCOAT, though similar in many ways to BHMIE, does not have as wide a range of applicability, and it should be used with more caution. The reasons for this will be discussed in the following paragraphs.

The mathematical form of all the scattering functions for a coated sphere—efficiencies and matrix elements—have the same form as those for a homogeneous sphere. Only the scattering coefficients (8.2) are different; these may be written in a form more suitable for computations:

$$a_n = \frac{\{\tilde{D}_n/m_2 + n/y\}\psi_n(y) - \psi_{n-1}(y)}{\{\tilde{D}_n/m_2 + n/y\}\xi_n(y) - \xi_{n-1}(y)},$$

$$b_n = \frac{\{m_2\tilde{G}_n + n/y\}\psi_n(y) - \psi_{n-1}(y)}{\{m_2\tilde{G}_n + n/y\}\xi_n(y) - \xi_{n-1}(y)},$$

$$\tilde{D}_n = \frac{D_n(m_2 y) - A_n\chi_n'(m_2 y)/\psi_n(m_2 y)}{1 - A_n\chi_n(m_2 y)/\psi_n(m_2 y)},$$

$$\tilde{G}_n = \frac{D_n(m_2 y) - B_n\chi_n'(m_2 y)/\psi_n(m_2 y)}{1 - B_n\chi_n(m_2 y)/\psi_n(m_2 y)},$$

$$A_n = \psi_n(m_2 x)\frac{mD_n(m_1 x) - D_n(m_2 x)}{mD_n(m_1 x)\chi_n(m_2 x) - \chi_n'(m_2 x)},$$

$$B_n = \psi_n(m_2 x)\frac{mD_n(m_2 x) - D_n(m_1 x)}{m\chi_n'(m_2 x) - D_n(m_1 x)\chi_n(m_2 x)}.$$

D_n is the logarithmic derivative ψ_n'/ψ_n and m is m_2/m_1. Because of the relation

$$\frac{1}{\psi_n} = \chi_n D_n - \chi_n',$$

which follows from the Wronskian (4.60), only three of the four functions ψ_n, χ_n, χ_n', D_n are independent. We also have

$$\chi_n'(z) = \chi_{n-1}(z) - \frac{n\chi_n(z)}{z}.$$

A_n and B_n depend on the size parameter x of the *inner*, or core, sphere only; they are independent of the thickness of the coating. Moreover, they are similar in form to the scattering coefficients for a homogeneous sphere with size parameter x. Therefore, it is reasonable to expect that we need not compute A_n and B_n for orders n appreciably larger than x. Convergence of the series for efficiencies and matrix elements, however, is determined by the size parameter y of the *outer* sphere. Therefore, if y is much greater than x, A_n and B_n will be computed well beyond the range where they are needed unless special care is taken. If this were merely inefficient, it would be tolerable. But, as we pointed out in Appendix A, one cannot expect to reliably compute by upward recurrence Bessel functions of order much larger than their argument. We have included, therefore, four tests in BHCOAT: if all the inequalities

$$\mathrm{DEL}|D_n(m_2y)| > |A_n\chi_n'(m_2y)/\psi_n(m_2y)| \quad \text{and} \quad |B_n\chi_n'(m_2y)/\psi_n(m_2y)|$$

$$\mathrm{DEL} > |A_n\chi_n(m_2y)/\psi_n(m_2y)| \quad \text{and} \quad |B_n\chi_n(m_2y)/\psi_n(m_2y)|$$

are satisfied for some index n, then $\tilde{D}_n$ and $\tilde{G}_n$ are set equal to $D_n(m_2y)$ for all successive indices; note that when this occurs, the scattering coefficients are identical with those for a homogeneous sphere with size parameter y and relative refractive index m_2. The inner sphere convergence criterion DEL is 10^{-8} in BHCOAT; it can, of course, be changed.

The fact that the contributions from the inner sphere to the scattering coefficients converge more rapidly than those associated with the outer sphere is merely a nuisance which is easily brushed aside. A more serious obstacle—one which we failed to hurdle—to writing an "explosion-proof" program for an *arbitrary* coated sphere is that the scattering coefficients cannot be put in a form that avoids computing excessively large numbers. The scattering coefficients for a homogeneous sphere can be written in such a way that functions of complex arguments are ratios—the logarithmic derivative—and the Riccati–Bessel functions have *real* arguments. This cannot be done with the coated sphere: we must, in general, compute functions $\psi_n(z)$ and $\chi_n(z)$ of the *complex* variable $z = z_r + iz_i$. That this can lead to difficulties is easily demonstrated. Consider the zero-order function with which upward recurrence is begun:

$$\chi_0(z) = \cos z = \frac{e^{z_i} + e^{-z_i}}{2}\cos z_r - i\frac{e^{z_i} - e^{-z_i}}{2}\sin z_r.$$

Because of the factor $\exp(z_i)$ it is always possible to generate numbers that exceed the limits of any computer if the particle is sufficiently large and absorbing. For a given computer it is difficult to set precise upper bounds on the imaginary parts of the arguments of the functions computed in BHCOAT: although the zero-order functions might not exceed the limits of the computer, successive higher-order functions computed by upward recurrence might. We recommend that BHCOAT not be used if any of the imaginary parts of $m_1 x$, $m_2 x$, or $m_2 y$ exceeds 30. This is only a rough guide, however; we strongly urge the user to do a bit of experimenting before accepting as correct any numbers produced by BHCOAT.

The convergence criterion in BHCOAT is the same as that in BHMIE: series are terminated after $y + 4y^{1/3} + 2$ terms. Unlike BHMIE, however, all functions, including logarithmic derivatives, are computed by upward recurrence: it seemed pointless to compute these derivatives by downward recurrence when they are not the major obstacle to writing a program valid for an arbitrary coated sphere.

Tests of BHCOAT We subjected BHCOAT to the same tests as BHMIE. In addition, BHCOAT agrees with BHMIE when $m_1 = m_2$. We found that the asymptotic limit (4.83) for the backscattering efficiency was also a sensitive indicator of the health of the coated sphere program. The composition and size of the core is irrelevant if almost all the incident light is absorbed in the coating. On *physical* grounds, therefore, we expect Q_b for large y to be approximately equal to the reflectance of a slab with refractive index m_2 if the coating is sufficiently thick and absorbing; this was verified by computations.

As a further test we compared efficiencies computed by BHCOAT with those given by Fenn and Oser (1965); there was good agreement in all instances.

```
      PROGRAM COAT (INPUT=TTY,OUTPUT=TTY,TAPE5=TTY)
C     ******************************************************************
C     COAT IS THE CALLING PROGRAM FOR BHCOAT, THE SUBROUTINE
C     THAT CALCULATES EFFICIENCIES FOR A COATED SPHERE.
C     FOR GIVEN RADII AND REFRACTIVE INDICES OF INNER AND
C     OUTER SPHERES, REFRACTIVE INDEX OF SURROUNDING
C     MEDIUM, AND FREE SPACE WAVELENGTH, COAT CALCULATES SIZE
C     PARAMETERS AND RELATIVE REFRACTIVE INDICES
C     ******************************************************************
C
C                   ***********CAUTION***********
C
C     BHCOAT SHOULD NOT BE USED FOR LARGE, HIGHLY ABSORBING
C     COATED SPHERES
C     X*REFIM1, X*REFIM2, AND Y*REFIM2 SHOULD BE LESS THAN ABOUT 30
C
C                   ***********CAUTION***********
C
C
      COMPLEX RFREL1,RFREL2
      WRITE (5,11)
C     ******************************************************************
C     REFMED = (REAL) REFRACTIVE INDEX OF SURROUNDING MEDIUM
C     ******************************************************************
      REFMED=1.0
C     ******************************************************
C     REFRACTIVE INDEX OF CORE = REFRE1 + I*REFIM1
C     REFRACTIVE INDEX OF COAT = REFRE2 + I*REFIM2
C     ******************************************************
      REFRE1=1.59
      REFIM1=.66
      REFRE2=1.409
      REFIM2=.1747
C     ************************************
C     RADCOR = RADIUS OF CORE
C     RADCOT = RADIUS OF COAT
C     RADCOR, RADCOT, WAVEL SAME UNITS
C     ************************************
      RADCOR=.171
      RADCOT=6.265
      WAVEL=3.
      WRITE (5,12) REFMED,REFRE1,REFIM1,REFRE2,REFIM2
      WRITE (5,13) RADCOR,RADCOT,WAVEL
      RFREL1=CMPLX(REFRE1,REFIM1)/REFMED
      RFREL2=CMPLX(REFRE2,REFIM2)/REFMED
      PI=3.14159265
      X=2.*PI*RADCOR*REFMED/WAVEL
      Y=2.*PI*RADCOT*REFMED/WAVEL
      WRITE (5,14) X,Y
      CALL BHCOAT (X,Y,RFREL1,RFREL2,QEXT,QSCA,QBACK)
      WRITE (5,67) QSCA,QEXT,QBACK
   11 FORMAT (/"COATED SPHERE SCATTERING PROGRAM"//)
   12 FORMAT (//5X,"REFMED = ",F8.4/5X,"REFRE1 =",E14.6,
     13X,"REFIM1 =",E14.6/5X,"REFRE2 =",E14.6,3X,"REFIM2 =",E14.6)
   13 FORMAT (5X,"CORE RADIUS =",F7.3,3X,"COAT RADIUS =",F7.3/
     15X,"WAVELENGTH =",F7.4)
   14 FORMAT (5X,"CORE SIZE PARAMETER = ",F8.3,3X,"COAT SIZE"
     1" PARAMETER =",F8.3)
   67 FORMAT (/,1X,"QSCA =",E13.6,3X,"QEXT =",E13.6,3X,
     1"QBACK =",E13.6//)
```

```
      STOP
      END
      SUBROUTINE BHCOAT (X,Y,RFREL1,RFREL2,QEXT,QSCA,QBACK)
C       ************************************************
C       SUBROUTINE BHCOAT CALCULATES EFFICIENCIES FOR
C       EXTINCTION, TOTAL SCATTERING, AND BACKSCATTERING
C       FOR GIVEN SIZE PARAMETERS OF CORE AND COAT AND
C       RELATIVE REFRACTIVE INDICES
C       ALL BESSEL FUNCTIONS COMPUTED BY UPWARD RECURRENCE
C       ************************************************
      COMPLEX RFREL1,RFREL2,X1,X2,Y2,REFREL
      COMPLEX D1X1,D0X1,D1X2,D0X2,D1Y2,D0Y2
      COMPLEX XI0Y,XI1Y,XIY,CHI0Y2,CHI1Y2,CHIY2,CHI0X2,CHI1X2,CHIX2
      COMPLEX CHIPX2,CHIPY2,ANCAP,BNCAP,DNBAR,GNBAR,AN,BN,CRACK,BRACK
      COMPLEX XBACK,AMESS1,AMESS2,AMESS3,AMESS4
      DEL=1.0E-8
C       *********************************************
C       DEL IS THE INNER SPHERE CONVERGENCE CRITERION
C       *********************************************
      X1=RFREL1*X
      X2=RFREL2*X
      Y2=RFREL2*Y
      YSTOP = Y + 4.*Y**.3333 + 2.
      REFREL=RFREL2/RFREL1
      NSTOP=YSTOP
C       ***********************************
C       SERIES TERMINATED AFTER NSTOP TERMS
C       ***********************************
      D0X1=CCOS(X1)/CSIN(X1)
      D0X2=CCOS(X2)/CSIN(X2)
      D0Y2=CCOS(Y2)/CSIN(Y2)
      PSI0Y=COS(Y)
      PSI1Y=SIN(Y)
      CHI0Y=-SIN(Y)
      CHI1Y=COS(Y)
      XI0Y=CMPLX(PSI0Y,-CHI0Y)
      XI1Y=CMPLX(PSI1Y,-CHI1Y)
      CHI0Y2=-CSIN(Y2)
      CHI1Y2=CCOS(Y2)
      CHI0X2=-CSIN(X2)
      CHI1X2=CCOS(X2)
      QSCA=0.0
      QEXT=0.0
      XBACK=CMPLX(0.0,0.0)
      N=1
      IFLAG=0
  200 RN=N
      PSIY=(2.*RN-1.)*PSI1Y/Y-PSI0Y
      CHIY=(2.*RN-1.)*CHI1Y/Y-CHI0Y
      XIY=CMPLX(PSIY,-CHIY)
      D1Y2=1./(RN/Y2-D0Y2)-RN/Y2
      IF (IFLAG.EQ.1) GO TO 999
      D1X1=1./(RN/X1-D0X1)-RN/X1
      D1X2=1./(RN/X2-D0X2)-RN/X2
      CHIX2=(2.*RN-1.)*CHI1X2/X2-CHI0X2
      CHIY2=(2.*RN-1.)*CHI1Y2/Y2-CHI0Y2
      CHIPX2=CHI1X2-RN*CHIX2/X2
      CHIPY2=CHI1Y2-RN*CHIY2/Y2
      ANCAP=REFREL*D1X1-D1X2
      ANCAP=ANCAP/(REFREL*D1X1*CHIX2-CHIPX2)
```

```
      ANCAP=ANCAP/(CHIX2*D1X2-CHIPX2)
      BRACK=ANCAP*(CHIY2*D1Y2-CHIPY2)
      BNCAP=REFREL*D1X2-D1X1
      BNCAP=BNCAP/(REFREL*CHIPX2-D1X1*CHIX2)
      BNCAP=BNCAP/(CHIX2*D1X2-CHIPX2)
      CRACK=BNCAP*(CHIY2*D1Y2-CHIPY2)
      AMESS1=BRACK*CHIPY2
      AMESS2=BRACK*CHIY2
      AMESS3=CRACK*CHIPY2
      AMESS4=CRACK*CHIY2
      IF(CABS(AMESS1).GT.DEL*CABS(D1Y2)) GO TO 999
      IF(CABS(AMESS2).GT.DEL) GO TO 999
      IF(CABS(AMESS3).GT.DEL*CABS(D1Y2)) GO TO 999
      IF(CABS(AMESS4).GT.DEL) GO TO 999
      BRACK=CMPLX(0.0,0.0)
      CRACK=CMPLX(0.0,0.0)
      IFLAG=1
  999 DNBAR=D1Y2-BRACK*CHIPY2
      DNBAR=DNBAR/(1.-BRACK*CHIY2)
      GNBAR=D1Y2-CRACK*CHIPY2
      GNBAR=GNBAR/(1.-CRACK*CHIY2)
      AN=(DNBAR/RFREL2+RN/Y)*PSIY-PSI1Y
      AN=AN/((DNBAR/RFREL2+RN/Y)*XIY-XI1Y)
      BN=(RFREL2*GNBAR+RN/Y)*PSIY-PSI1Y
      BN=BN/((RFREL2*GNBAR+RN/Y)*XIY-XI1Y)
      QSCA=QSCA+(2.*RN+1.)*(CABS(AN)*CABS(AN)+CABS(BN)*CABS(BN))
      XBACK=XBACK+(2.*RN+1.)*(-1.)**N*(AN-BN)
      QEXT=QEXT+(2.*RN+1.)*(REAL(AN)+REAL(BN))
      PSI0Y=PSI1Y
      PSI1Y=PSIY
      CHI0Y=CHI1Y
      CHI1Y=CHIY
      XI1Y=CMPLX(PSI1Y,-CHI1Y)
      CHI0X2=CHI1X2
      CHI1X2=CHIX2
      CHI0Y2=CHI1Y2
      CHI1Y2=CHIY2
      D0X1=D1X1
      D0X2=D1X2
      D0Y2=D1Y2
      N=N+1
      IF(N-1-NSTOP) 200,300,300
  300 QSCA=(2./(Y*Y))*QSCA
      QEXT=(2./(Y*Y))*QEXT
      QBACK=XBACK*CONJG(XBACK)
      QBACK=(1./(Y*Y))*QBACK
      RETURN
      END
```

```
SHELL

COATED SPHERE SCATTERING PROGRAM



     REFMED =    1.0000
     REFRE1 =   .159000E+01   REFIM1 =    .660000E+00
     REFRE2 =   .140900E+01   REFIM2 =    .174700E+00
     CORE RADIUS =   .171   COAT RADIUS =  6.265
     WAVELENGTH = 3.0000
     CORE SIZE PARAMETER =     .358   COAT SIZE PARAMETER =  13.121

 QSCA =  .114341E+01   QEXT =  .232803E+01   QBACK =  .285099E-01
```