Absorption and Scattering of Light by Small Particles
CRAIG F. BOHREN, DONALD R. HUFFMAN
© 2004 WILEY-VCH Verlag GmbH & Co.

Appendix C

Normally Illuminated Infinite Cylinder

The program described in this appendix is an improved version of one contained in a report by Bohren and Timbrell (1979); the underlying theory and definitions of symbols are given in Section 8.4.

The coefficients (8.38) are still not in a form suitable for computations: indices are required to be positive integers in most versions of Fortran. We can get around this easily enough by defining $G_n = D_{n-1}$ $(n = 1, 2, \dots)$, which satisfies the recurrence relation

$$G_{n-1}(z) = \frac{n-2}{z} - \frac{1}{\dfrac{n-1}{z} + G_n(z)}. \tag{C.1}$$

If we use the relations $J_{-n} = (-1)^n J_n$ and $Y_{-n} = (-1)^n Y_n$, the coefficients a_0 and b_0 may be written

$$a_0 = \frac{G_1(mx)BJ_1(x)/m + BJ_2(x)}{G_1(mx)BH_1(x)/m + BH_2(x)}, \qquad b_0 = \frac{mG_1(mx)BJ_1(x) + BJ_2(x)}{mG_1(mx)BH_1(x) + BH_2(x)},$$

where we define $BJ_n = J_{n-1}$ and $BH_n = H_{n-1}^{(1)}$ $(n = 1, 2, \dots)$. For $n \neq 0$ we have

$$a_n = \frac{\{G_{n+1}(mx)/m + n/x\}BJ_{n+1}(x) - BJ_n(x)}{\{G_{n+1}(mx)/m + n/x\}BH_{n+1}(x) - BH_n(x)},$$

$$b_n = \frac{\{mG_{n+1}(mx) + n/x\}BJ_{n+1}(x) - BJ_n(x)}{\{mG_{n+1}(mx) + n/x\}BH_{n+1}(x) - BH_n(x)}.$$

As in the previous programs, series for scattering matrix elements and efficiencies are truncated after NSTOP terms, where NSTOP $= x + 4x^{1/3} + 2$. $G_n(mx)$ is computed by (C.1): beginning with G_{NMX}, successive lower-order logarithmic derivatives $G_{\mathrm{NMX}-1}, \dots, G_1$ are computed by downward recurrence. Provided that NMX is sufficiently greater than NSTOP and $|mx|$, G_p for

$p \leqslant$ NSTOP is insensitive to the choice of G_{NMX}. We varied G_{NMX} over five orders of magnitude for a range of arguments mx; in each instance $G_{\mathrm{NMX}-5}$ was independent of G_{NMX}. Therefore, we take NMX = Max(NSTOP,$|mx|$) + 15, and recurrence begins with $G_{\mathrm{NMX}} = 0.0 + i0.0$.

Computation of Bessel Functions The Bessel functions J_n and Y_n pose more computational problems than the logarithmic derivative. In BHCYL these functions are computed by an algorithm credited to Miller (British Association, 1952, p. xvii), further details of which are given by Stegun and Abramowitz (1957) and by Goldstein and Thaler (1959); we outline this scheme in the following paragraph.

To calculate $J_n(x)$ we generate a sequence of functions $F_p(x)$—in BHCYL $F(P)$ is F_{p-1}—by *downward* recurrence:

$$F_{p-1}(x) = \frac{2pF_p(x)}{x} - F_{p+1}(x) \qquad (p = M-1, \ldots, 1) \tag{C.2}$$

beginning with $F_M = 0.0$ and $F_{M-1} = 10^{-32}$, where M is greater than both n and x. Although F_{M-1} is arbitrary, it should be small to avoid generating excessively large numbers when $x < 1$. J_p and Y_p also satisfy (C.2), and for all p sufficiently smaller than M we have

$$J_p(x) \simeq \frac{F_p(x)}{\alpha},$$

where α is independent of p; this constant is evaluated by using

$$J_0 + 2\sum_{m=1}^{\infty} J_{2m} = 1,$$

from which it follows that

$$\alpha \simeq F_0(x) + 2\sum_{m=1}^{\frac{1}{2}M-1} F_{2m}(x).$$

$Y_n(x)$ is computed by *upward* recurrence

$$Y_{p+1}(x) = \frac{2pY_p(x)}{x} - Y_{p-1}(x),$$

beginning with Y_0 and Y_1; the first of these is obtained from

$$Y_0(x) = \frac{2}{\pi}\left[\left(\log\frac{x}{2} + \gamma\right)J_0(x) - 2\sum_{m=1}^{\infty}\frac{(-1)^m J_{2m}(x)}{m}\right],$$

where Euler's constant γ is 0.57721566...; the second is calculated from the Wronskian

$$J_1(x)Y_0(x) - J_0(x)Y_1(x) = \frac{2}{\pi x}.$$

The choice of M = MST in the recurrence relation (C.2) is sometimes determined by iteration. This may be appropriate in a program to compute $J_n(x)$ for arbitrary n and x. But our needs are more limited, and we can avoid iterating—thereby designing a more efficient program—by properly choosing MST. NSTOP, which is greater than x, is the largest order of the Bessel functions required in scattering calculations. Therefore, we take MST to be the even integer closest to NSTOP + NDELTA. To determine NDELTA we did a series of calculations: NDELTA was incremented until successive values of J_{NSTOP} agreed to nine decimal places; the size parameter was varied from 0.001 to 5000. Our results are summarized by the empirical relation

$$\mathrm{NDELTA} = (101 + x)^{0.499}.$$

Tests of BHCYL Computed values of $J_n(x)$ and $Y_n(x)$ for various n and x were compared with values tabulated by Olver (1964). Computed logarithmic derivatives were also compared with values calculated from tabulated Bessel functions. In all instances there was agreement to as many decimal places as were given in the tables. This gives us some confidence that BHCYL does what it was designed to do.

Scattering coefficients for absorbing cylinders over a limited range of m and x have been tabulated by Libelo (1962). These were compared with coefficients computed by BHCYL; in all instances there was agreement to as many decimal places as were given in the tables.

There are a few obvious checks on the scattering and extinction efficiencies: (1) they are nonnegative for all x and m; and (2) Q_{sca} and Q_{ext} are equal when the cylinder is nonabsorbing. BHCYL has never failed to pass these tests. Scattering efficiencies are calculated in BHCYL as the sum of squares of the scattering coefficients, whereas extinction efficiencies are calculated from the real part of the forward amplitude scattering matrix elements T_1 and T_2. This was done purposely to provide a check on the amplitude scattering matrix elements.

As a further check, we compared efficiencies computed by BHCYL with values given by Larkin and Churchill (1959); in all instances, there was good agreement.

```
      PROGRAM CALCYL (INPUT=TTY,OUTPUT=TTY,TAPE5=TTY)
C       ***********************************************************
C       CALCYL CALCULATES THE SIZE PARAMETER (X) AND RELATIVE
C       REFRACTIVE INDEX (REFREL) FOR A GIVEN CYLINDER REFRACTIVE
C       INDEX, MEDIUM REFRACTIVE INDEX, RADIUS, AND FREE SPACE
C       WAVELENGTH. IT THEN CALLS BHCYL, THE SUBROUTINE THAT COMPUTES
C       AMPLITUDE SCATTERING MATRIX ELEMENTS AND EFFICIENCIES
C       ***********************************************************
      COMPLEX REFREL,T1(200),T2(200)
      DIMENSION ANG(200)
      WRITE (5,11)
C       **********************************************************
C       REFMED = (REAL) REFRACTIVE INDEX OF SURROUNDING MEDIUM
C       **********************************************************
      REFMED=1.0
C       ***********************************************
C       REFRACTIVE INDEX OF CYLINDER = REFRE + I*REFIM
C       **********************************************
      REFRE=1.55
      REFIM=0.0
      REFREL=CMPLX(REFRE,REFIM)/REFMED
      WRITE (5,12) REFMED,REFRE,REFIM
      PI=3.14159265
C       ********************************
C       RADIUS (RAD) AND WAVELENGTH (WAVEL) SAME UNITS
C       ********************************
      RAD=.525
      WAVEL=.6328
      X=2.*PI*RAD*REFMED/WAVEL
      WRITE (5,13) RAD,WAVEL
      WRITE (5,14) X
C       *********************************************
C       FIN = FINAL ANGLE (DEGREES)
C       INTANG = NUMBER OF INTERVALS BETWEEN 0 AND FIN
C       *********************************************
      FIN=180.
      INTANG=20
      WRITE (5,15)
      CALL BHCYL (X,REFREL,T1,T2,QSCPAR,QSCPER,QEXPAR,QEXPER,
     1FIN,INTANG,ANG)
      NPTS=INTANG+1
      T11NOR=0.5*(CABS(T1(1))*CABS(T1(1)))
      T11NOR=T11NOR+0.5*(CABS(T2(1))*CABS(T2(1)))
C       ************************************************************
C       T33 AND T34 MATRIX ELEMENTS NORMALZIED BY T11
C       T11 IS NORMALIZED TO 1.0 IN THE FORWARD DIRECTION
C       POL = DEGREE OF POLARIZATION (INCIDENT UNPOLARIZED LIGHT)
C       ************************************************************
      DO 107 J=1,NPTS
      TPAR=CABS(T1(J))
      TPAR=TPAR*TPAR
      TPER=CABS(T2(J))
      TPER=TPER*TPER
      T11=0.5*(TPAR+TPER)
      T12=0.5*(TPAR-TPER)
      POL=T12/T11
      T33=REAL(T1(J)*CONJG(T2(J)))
      T34=AIMAG(T1(J)*CONJG(T2(J)))
      T33=T33/T11
      T34=T34/T11
```

```
      T11=T11/T11NOR
  107 WRITE (5,68) ANG(J),T11,POL,T33,T34
      WRITE  (5,67)  QSCPAR,QEXPAR,QSCPER,QEXPER
   67 FORMAT (//,"QSCPAR =",E14.6,3X,"QEXPAR =",E14.6/
     1"QSCPER =",E14.6,3X,"QEXPER =",E14.6//)
   68 FORMAT (1X,F8.2,2X,E13.6,2X,E13.6,2X,E13.6,2X,E13.6)
   11 FORMAT (/"CYLINDER PROGRAM: NORMALLY INCIDENT LIGHT"//)
   12 FORMAT (5X,"REFMED =",F8.4,3X,"REFRE =",E14.6,3X
     1"REFIM =",E14.6)
   13 FORMAT (5X,"CYLINDER RADIUS =",F7.3,3X,"WAVELENGTH =",F7.4)
   14 FORMAT (5X,"SIZE PARAMETER =",F8.3/)
   15 FORMAT (//2X,"ANGLE",7X,"T11",13X,"POL",13X,"T33",13X,"T34"//)
      STOP
      END
C        ***********************************************************
C        SUBROUTINE BHCYL CALCULATES AMPLITUDE SCATTERING MATRIX
C        ELEMENTS AND EFFICIENCIES FOR EXTINCTION AND SCATTERING
C        FOR A GIVEN SIZE PARAMETER AND RELATIVE REFRACTIVE INDEX
C        THE INCIDENT LIGHT IS NORMAL TO THE CYLINDER AXIS
C        PAR:ELECTRIC FIELD PARALLEL TO CYLINDER AXIS
C        PER:ELECTRIC FIELD PERPENDICULAR TO CYLINDER AXIS
C        ***********************************************************
      SUBROUTINE BHCYL (X,REFREL,T1,T2,QSCPAR,QSCPER,QEXPAR,QEXPER,
     2FIN,INTANG,ANG)
      COMPLEX REFREL,Y,AN,BN,A0,B0
      COMPLEX G(1000),BH(1000),T1(200),T2(200)
      DIMENSION THETA(200),ANG(200),BJ(1000),BY(1000),F(1000)
      Y=X*REFREL
      XSTOP=X+4.*X**.3333+2.
C        ***********************************
C        SERIES TERMINATED AFTER NSTOP TERMS
C        ***********************************
      NSTOP=XSTOP
      YMOD=CABS(Y)
      NMX=AMAX1(XSTOP,YMOD)+15
      NPTS=INTANG+1
      DANG=FIN/FLOAT(INTANG)
      DO 555 J=1,NPTS
      ANG(J)=(FLOAT(J)-1.)*DANG
  555 THETA(J)=ANG(J)*0.017453292
C        ******************************************************
C        LOGARITHMIC DERIVATIVE G(J) CALCULATED BY DOWNWARD
C        RECURRENCE BEGINNING WITH INITIAL VALUE 0.0 + I*0.0
C        AT J = NMX
C        ******************************************************
      G(NMX)=CMPLX(0.0,0.0)
      NN=NMX-1
      DO 120 N=1,NN
      RN=NMX-N+1
      K=NMX-N
  120 G(K)=((RN-2.)/Y)-(1./(G(K+1)+(RN-1.)/Y))
C        ******************************************************
C        BESSEL FUNCTIONS J(N) COMPUTED BY DOWNWARD RECURRENCE
C        BEGINNING AT N □ NSTOP + NDELTA
C        BESSEL FUNCTIONS Y(N) COMPUTED BY UPWARD RECURRENCE
C        BJ(N+1) = J(N),  BY(N+1) = Y(N)
C        ******************************************************
      NDELTA=(101.+X)**.499
      MST=NSTOP+NDELTA
      MST=(MST/2)*2
```

```
      F(MST+1)=0.0
      F(MST)=1.0E-32
      M1=MST-1
      DO 201 L=1,M1
      ML=MST-L
  201 F(ML)=2.*FLOAT(ML)*F(ML+1)/X-F(ML+2)
      ALPHA=F(1)
      M2=MST-2
      DO 202 LL=2,M2,2
  202 ALPHA=ALPHA+2.*F(LL+1)
      M3=M2+1
      DO 203 N=1,M3
  203 BJ(N)=F(N)/ALPHA
      BY(1)=BJ(1)*(ALOG(X/2.)+.577215664)
      M4=MST/2-1
      DO 204 L=1,M4
  204 BY(1)=BY(1)-2.*((-1.)**L)*BJ(2*L+1)/FLOAT(L)
      BY(1)=.636619772*BY(1)
      BY(2)=BJ(2)*BY(1)-.636619772/X
      BY(2)=BY(2)/BJ(1)
      NS=NSTOP-1
      DO 205 KK=1,NS
  205 BY(KK+2)=2*FLOAT(KK)*BY(KK+1)/X-BY(KK)
      NN=NSTOP+1
      DO 715 N=1,NN
  715 BH(N)=CMPLX(BJ(N),BY(N))
      A0=G(1)*BJ(1)/REFREL+BJ(2)
      A0=A0/(G(1)*BH(1)/REFREL+BH(2))
      B0=REFREL*G(1)*BJ(1)+BJ(2)
      B0=B0/(REFREL*G(1)*BH(1)+BH(2))
      QSCPAR=CABS(B0)*CABS(B0)
      QSCPER=CABS(A0)*CABS(A0)
      DO 101 K=1,NPTS
      T1(K)=B0
  101 T2(K)=A0
      DO 123 N=1,NSTOP
      RN=N
      AN=(G(N+1)/REFREL+RN/X)*BJ(N+1)-BJ(N)
      AN=AN/((G(N+1)/REFREL+RN/X)*BH(N+1)-BH(N))
      BN=(REFREL*G(N+1)+RN/X)*BJ(N+1)-BJ(N)
      BN=BN/((REFREL*G(N+1)+RN/X)*BH(N+1)-BH(N))
      DO 102 J=1,NPTS
      C=COS(RN*THETA(J))
      T1(J)=2.*BN*C+T1(J)
  102 T2(J)=2.*AN*C+T2(J)
      QSCPAR=QSCPAR+2.*CABS(BN)*CABS(BN)
  123 QSCPER=QSCPER+2.*CABS(AN)*CABS(AN)
      QSCPAR=(2./X)*QSCPAR
      QSCPER=(2./X)*QSCPER
      QEXPER=(2./X)*REAL(T2(1))
      QEXPAR=(2./X)*REAL(T1(1))
      RETURN
      END
```

CYL

CYLINDER PROGRAM: NORMALLY INCIDENT LIGHT

REFMED = 1.0000 REFRE = .155000E+01 REFIM = 0.
CYLINDER RADIUS = .525 WAVELENGTH = .6328
SIZE PARAMETER = 5.213

ANGLE	T11	POL	T33	T34
0.00	.100000E+01	.734486E-01	.997149E+00	-.172894E-01
9.00	.686631E+00	.291477E-01	.999432E+00	-.169025E-01
18.00	.217683E+00	-.135736E+00	.986700E+00	.894351E-01
27.00	.144205E+00	.103749E+00	.931604E+00	.348352E+00
36.00	.259646E+00	.162651E+00	.977440E+00	.134744E+00
45.00	.231162E+00	-.894687E-02	.997329E+00	.724853E-01
54.00	.132150E+00	-.179789E+00	.953175E+00	.243175E+00
63.00	.839900E-01	-.349048E-01	.900228E+00	.434018E+00
72.00	.669177E-01	.504876E-01	.937414E+00	.344536E+00
81.00	.622477E-01	-.823535E-02	.942424E+00	.334320E+00
90.00	.482920E-01	-.510106E-01	.967653E+00	.247076E+00
99.00	.199993E-01	-.606254E+00	.782214E+00	.143519E+00
108.00	.244164E-01	-.141679E+00	.173427E+00	.974602E+00
117.00	.416869E-01	.476291E+00	.534335E+00	.698307E+00
126.00	.200601E-01	.488882E+00	.839228E+00	-.238100E+00
135.00	.186030E-01	-.671603E+00	-.708250E+00	.217558E+00
144.00	.655546E-01	-.676521E-01	-.325732E+00	.943039E+00
153.00	.632725E-01	.262420E-01	-.223743E+00	.974295E+00
162.00	.168029E-01	-.282769E-01	-.771987E+00	.635010E+00
171.00	.333764E-01	.956354E+00	-.135136E+00	.259084E+00
180.00	.673014E-01	.899741E+00	.641930E-01	.431676E+00

QSCPAR = .209716E+01 QEXPAR = .209716E+01
QSCPER = .192782E+01 QEXPER = .192782E+01